The *Contrasts.sps* and *Partitioning Interaction.sps* files provide analyses not readily available from SPSS's *Univariate* menu. They are intended for analyses following the ANOVAs of data from the between-subject designs of Chapters 8 - 12. The file *Nested Designs.sps* provides an analysis of data from such designs with the flexibility to deal with various combinations of variables that cross or are nested within each other, and that have fixed or random effects. The file *Nested Designs_2.sps* extends this by including a between-subjects factor. Both of these files calculate Quasi_*F* statistics and the appropriate degrees of freedom for their error terms. The book also references websites containing software that can be downloaded at little or no cost that provide confidence intervals for both raw and standardized effects.
The files have been designed to run with SPSS data files on the website. However, by changing or adding variable names and contrast weights, and by modifying commands, the files can be used with your own data sets. In general,

1. Open the data set in SPSS.
2. Open the syntax file.
3. Select from the syntax file menu *Run*, then *All*.

Step 1 is not necessary if the first statement in your syntax file is a *GET FILE* statement; e.g.

GET FILE='Table 12_1 IA2 Data.sav'.

Following is a description of each of the syntax files on the website.

Contrasting Group Means

The *Contrasts* option in the *General Linear Model/Unianova* menu provides several different contrasts including polynomial analyses. However, even greater flexibility can be achieved with a syntax file. The syntax file, Contrasts.sps, was written to be run with the SPSS data file *Table 12_1 IA2 Data.Sav*. Results are for the difference between the mean of the Attribute condition and the mean of the combined Name and Random conditions, and the difference between the means of the Name and Random conditions. The output contains the ANOVA results and otherwise is the same as in Table 12.4 of the book. The syntax is:

```
 UNIANOVA
Recall BY Organization Delay
/CONTRAST (Organization)= Special(1 -.5 -.5, 0 1 -1)
/METHOD = SSTYPE(3)
/INTERCEPT = INCLUDE
/CRITERIA = ALPHA(.05)
/DESIGN = Organization Delay Organization*Delay.
```

As stated previously, variable names and contrast weights can be changed or added to fit your study. Note that the syntax ends with the DESIGN statement. Also note that this statement must end with a period.

Analyzing Interactions

In Chapter 11, we presented data for a two-factor between subjects design; one factor was
clinical category and the second was stimulus height GSR was plotted in Figure 11.3 of the book as a function of stimulus height for each of the three clinical categories. Suppose we wish to

compare the slope of the Control curve with the average slope of the clinical curves; call this *Linear C vs MS, SS* . The syntax file *Partitioning Interaction.sps* provides this contrast plus several others. As an example of the basis for the coefficients, consider the *Linear (C vs MS, SS)* contrast. Let w_k be the linear coefficient for the *k*th stimulus height. Then the null hypothesis is

$$\Sigma w_k \overline{Y}_{k,C} - (1/2)(\Sigma w_k \overline{Y}_{k,MS} + \Sigma\, w_k \overline{Y}_{k,SS}) = 0$$

In order to deal with integers, we can multiply the weights by 2 so that the null hypothesis is

$$(2)\Sigma w_k \overline{Y}_{k,C} - (\Sigma w_k \overline{Y}_{k,MS} + \Sigma\, w_k \overline{Y}_{k,SS}) = 0$$

When there are five equally spaced levels, the linear coefficients in Appendix Table C6 are -2, -1, 0, 1, and 2. The weights in the first contrast in the syntax file now follow from our statement of the null hypothesis. The syntax which provides several contrasts is

```
UNIANOVA GSR BY Category Stimulus
 /LMATRIX = "Linear (C vs MS, SS)"
    Category * Stimulus -4 -2 0 2 4 2 1 0 -1 -2 2 1 0 -1 -2
 /LMATRIX = "Linear (MS vs SS)"
   Category * Stimulus 0 0 0 0 0 2 1 0 -1 -2 -2 -1 0 1 2
 /LMATRIX = "Quadratic (C vs MS, SS)"
   Category * Stimulus 4 -2 -4 -2 4 -2 1 2 1 -2 -2 1 2 1 -2
 /LMATRIX = "Quadratic (MS vs SS)"
   Category * Stimulus 0 0 0 0 0 -2 1 2 1 -2 2 -1 -2 -1 2
 /DESIGN = Category, stimulus, Category * Stimulus.
```

We could have also included contrasts of the cubic and quartic components of the curves. And contrasts other than polynomial ones can be analyzed with the proper choice of weights. Whatever contrasts you test, note our use of labels in quotation marks. Without such labels, the output can be confusing.

We opened the *GSR2 Data.sav* file and ran *Partitioning Interaction.sps*. The output includes results in a form similar to that of Table 12.4 for each contrast; e.g., the output for the linear contrast of the control curve vs the average of the two schizophrenic curves contains

Contrast Results (K Matrix)[a]

Contrast			Dependent Variable GSR
L1	Contrast Estimate		4.400
	Hypothesized Value		0
	Difference (Estimate - Hypothesized)		4.400
	Std. Error		3.444
	Sig.		.205
	95% Confidence Interval for Difference	Lower Bound	-2.460
		Upper Bound	11.260

a. Based on the user-specified contrast coefficients (L') matrix: Linear (C vs MS, SS)

An ANOVA table is also provided for each contrast. For the two linear contrasts, we have

Test Results

Dependent Variable:GSR

Source	Sum of Squares	df	Mean Square	F	Sig.
Contrast	.056	1	.056	.048	.828
Error	88.943	75	1.186		

and

Test Results

Dependent Variable:GSR

Source	Sum of Squares	df	Mean Square	F	Sig.
Contrast	1.936	1	1.936	1.633	.205
Error	88.943	75	1.186		

The researcher may wish to test whether there is *any* significant variation among the three
slopes. Because the two linear contrasts form a complete orthogonal set, we can calculate from the preceding two ANOVA tables $F = [(1.936 + .056)/2]/ 1.186 = .940$. This F is distributed on 2 and 75 *df.* It is clearly not significant.

Nesting, Random Effects, and Quasi-F Tests

The CD includes two examples of SPSS syntax files designed to analyze data from repeated measures designs in which variables may cross or be nested, and have fixed or random effects. The input to these files is not in the usual repeated-measures format. Instead, each independent variable is represented by a column as in analyses of between-subjects design. For example, the file *Table 16_1 uni.sav* represents the data in *Table 16_1 Nested Data.sav* in this univariate format. The file looks like



	S	A	B	Y	var	var
1	1	1	1	1584		
2	1	1	2	1563		
3	1	1	3	992		
4	1	1	4	1404		
5	1	1	5	754		
6	1	2	1	1273		
7	1	2	2	1821		
8	1	2	3	1507		
9	1	2	4	1321		
10	1	2	5	1701		
11	1	3	1	1636		
12	1	3	2	2152		
13	1	3	3	2051		
14	1	3	4	2070		
15	1	3	5	1584		
16	2	1	1	776		
17	2	1	2	1067		

To treat B (items) as having random effects and nested within levels of A, we use the SPSS syntax file *Nested Designs.sps:*

```
GLM Y BY A S B
 /RANDOM=S B
 /METHOD=SSTYPE(3)
 /INTERCEPT=INCLUDE
 /CRITERIA=ALPHA(0.05)
 /DESIGN=A S B(A) A BY S.
```

Note that the residual error term (*SxB/A*) is not included in the DESIGN statement. If *B* has fixed effects, it should be removed from the *RANDOM* statement. If items are not nested in levels of *A*, the *DESIGN* statement should be

```
/DESIGN= A S B A*S B*S A*B A*B*S.
```

(A*S and A BY S have the same meaning, as do B(A) and B WITHIN A.)

The SPSS output for *Table 16_1 uni.sav* and *Nested Designs.sps* is[1]

Tests of Between-Subjects Effects

Dependent Variable:Y

Source		Type III Sum of Squares	df	Mean Square	F	Sig.
Intercept	Hypothesis	2.767E8	1	2.767E8	148.641	.000
	Error	1.651E7	8.866	1.862E6[a]		
A	Hypothesis	2147716.467	2	1073858.233	3.375	.060
	Error	5052466.979	15.878	318201.510[b]		
S	Hypothesis	1.150E7	7	1642265.646	16.645	.000
	Error	1381266.200	14	98661.871[c]		
B(A)	Hypothesis	3127760.500	12	260646.708	6.341	.000
	Error	3452993.900	84	41107.070[d]		
A * S	Hypothesis	1381266.200	14	98661.871	2.400	.007
	Error	3452993.900	84	41107.070[d]		

a. MS(S) + MS(B(A)) - MS(Error)
b. MS(B(A)) + MS(A * S) - MS(Error)
c. MS(A * S)
d. MS(Error)

Note that the results are the same as in Table 16.2 and Box 16.1. In particular, the program computes the quasi-*F* ratio with the same value of *F* and denominator *df* that is present in Box 16.1.

The syntax file *Nested Designs_2.sps* expands the preceding file to include a between-subjects factor. The input in this example is Table 17_1 *Uni.Sav.* The syntax file is

```
UNIANOVA
Y BY Test Duration Subject Item
/RANDOM = Subject Item
/CONTRAST (Duration)=Polynomial
/METHOD = SSTYPE(3)
/INTERCEPT = INCLUDE
/PLOT = PROFILE( Duration*Test )
/CRITERIA = ALPHA(.05)
/DESIGN = Test Duration Subject(Test) Item(Duration) Test*Subject(Duration) Test*Duration
Test*Item(Duration).
```

Test is the between-subjects variable. There are different items in each duration. Once again the format for the input data file is as if all variables were between subjects; there are separate columns for subjects, test, duration, items, and Y. The output is

[1]There is a difference of opinion among statisticians about the status of interactions involving a fixed and a random factor (e.g., AxS) in constructing expected mean squares. The approach in this book follows that of Tukey and Cornfield (1956), and is consistent with that of many other textbooks, but differs from the approach taken in SPSS. We do not test *S* against *SA*. However, the quasi_*F* tests and their error *df* are the same in both approaches for the designs we have considered.

Tests of Between-Subjects Effects

Dependent Variable:Y

Source		Type III Sum of Squares	df	Mean Square	F	Sig.
Intercept	Hypothesis	974792.469	1	974792.469	2480.781	.000
	Error	8820.952	22.449	392.938[a]		
Test	Hypothesis	664.225	1	664.225	1.767	.198
	Error	7820.311	20.806	375.863[b]		
Duration	Hypothesis	4986.672	2	2493.336	72.325	.000
	Error	273.527	7.934	34.474[c]		
Subject(Test)	Hypothesis	8474.306	22	385.196	14.409	.000
	Error	1176.211	44	26.732[d]		
Item(Duration)	Hypothesis	473.483	12	39.457	1.763	.170
	Error	268.583	12	22.382[e]		
Test * Subject(Duration)	Hypothesis	1176.211	44	26.732	.843	.749
	Error	8372.733	264	31.715[f]		
Test * Duration	Hypothesis	72.317	2	36.158	2.078	.222
	Error	85.234	4.899	17.399[g]		
Test * Item(Duration)	Hypothesis	268.583	12	22.382	.706	.746
	Error	8372.733	264	31.715[f]		

a. MS(Subject(Test)) + MS(Item(Duration)) - MS(Error)

b. MS(Subject(Test)) + MS(Test * Item(Duration)) - MS(Error)

c. MS(Item(Duration)) + MS(Test * Subject(Duration)) - MS(Error)

d. MS(Test * Subject(Duration))

e. MS(Test * Item(Duration))

f. MS(Error)

g. MS(Test * Subject(Duration)) + MS(Test * Item(Duration)) - MS(Error)

Analyses Involving Categorical and Continuous Variables

In Appendix 24.1 we illustrated the use of regression analysis to test the interaction of a categorical variable (educational level, EL) and a continuous variable (age) with total cholesterol (tc) as the dependent variable. With the data file *Age_BMI.sav* open in SPSS, the syntax file EL_by_age.sps first tests the effects of EL and then the interaction effects:

```
RECODE schoolyr (Lowest thru 3=1) (Lowest thru 4=2) (Lowest thru 5=2) (Lowest thru 6=2) (Lowest thru 7=3)
(Lowest thru 8=3) INTO EL.
EXECUTE.
UNIANOVA tc BY EL
  /METHOD=SSTYPE(3)
  /INTERCEPT=INCLUDE
  /CRITERIA=ALPHA(0.05)
  /DESIGN=EL.
UNIANOVA tc BY EL WITH age
  /METHOD=SSTYPE(3)
  /INTERCEPT=INCLUDE
  /CRITERIA=ALPHA(0.05)
  /DESIGN=EL age EL*age.
```

Agsain note that each DESIGN statement must end with a period.